

Lab Exercise 5- Creating Formula Fields on a Data Lake Object (DLO)

Objective:

Learn how to create and validate formula fields in Salesforce Data Cloud to perform calculations, conditional logic, and data transformation at the DLO level.

1. Lab Overview

Step	Task	Concept Learned
1	Create and upload a sample CSV dataset	DLO creation and ingestion
2	Create basic formula fields	Arithmetic and derived metrics
3	Create conditional and text formulas	IF, CONCAT, TEXT functions
4	Validate formula outputs	Using Data Explorer
5	Apply real-world logic	Region category, commission, discount

2. Dataset Setup

Create a simple CSV file named **Purchase_Transactions.csv** with the following data:

transaction_id	customer_id	purchase_date	amount	region	payment_mode
T1001	C001	2025-10-10	850	East	Credit Card
T1002	C002	2025-10-12	450	West	Debit Card
T1003	C001	2025-10-15	25	East	(blank)
T1004	C003	2025-10-18	1200	North	NetBanking
T1005	C004	2025-10-19	300	South	UPI

Upload this file into **Salesforce Data Cloud → Data Manager → Data Streams**

to create a DLO named:

Purchase_Transactions_DLO

3. Creating Formula Fields in the DLO Schema

Formula 1: Discount Calculation

Objective: Apply a 10% discount to each transaction amount.

Formula:

amount * 0.9

New Field Name: discounted_amount__c

Data Type: Number

Expected Result:

amount	discounted_amount
850	765
450	405
25	22.5

Formula 2: Transaction Type (Conditional Logic)

Objective: Identify high-value transactions (> ₹500).

Formula:

IF(amount > 500, "High Value", "Regular")

New Field Name: transaction_type__c

Data Type: Text

Expected Result:

amount	transaction_type
850	High Value
450	Regular
25	Regular
1200	High Value
300	Regular

Formula 3: Extract Purchase Month

Objective: Get the month number from the purchase date.

Formula:

```
TEXT(MONTH(purchase_date))
```

New Field Name: purchase_month__c

Data Type: Text

Expected Result:

purchase_date	purchase_month
2025-10-10	10
2025-10-12	10

Formula 4: Combine Customer and Transaction IDs

Objective: Create a unique transaction label for reporting.

Formula:

```
customer_id & "-" & transaction_id
```

New Field Name: transaction_label__c

Data Type: Text

Expected Result:

customer_id	transaction_id	transaction_label
C001	T1001	C001-T1001
C002	T1002	C002-T1002

Formula 5: Data Cleansing for Payment Mode

Objective: Replace null or blank payment modes with “Unknown.”

Formula:

IF(ISBLANK(payment_mode), "Unknown", payment_mode)
--

New Field Name: cleaned_payment_mode__c

Data Type: Text

Expected Result:

payment_mode	cleaned_payment_mode
Credit Card	Credit Card
(blank)	Unknown

4. Advanced (Optional) Formula Fields

Formula 6: Commission Calculation

Objective: Calculate 5% commission for high-value transactions (> ₹500).

Formula:

IF(amount > 500, amount * 0.05, 0)

New Field Name: commission_amount__c

Data Type: Number

Formula 7: Region Category

Objective: Group regions into business zones.

Formula:

```
IF(
  OR(region = "East", region = "West"), "Zone 1",
  IF(OR(region = "North", region = "South"), "Zone 2", "Other")
)
```

New Field Name: region_category__c

Data Type: Text

5. Validation Steps in Data Cloud

1. Go to **Data Explorer** → Select **Purchase_Transactions_DLO**
 2. Click **Preview Data**
 3. Verify that:
 - Each formula field appears in the schema
 - Computed values display correctly for sample rows
 4. Use **Filter** or **Sort** to confirm formulas (e.g., all high-value > ₹500 are flagged)
-